C3M6

Extrema of Functions of Two Variables

In this assignment we will be solving simultaneous equations and using matrices in order to locate possible relative maxima and minima. After defining the function f and determining the partial derivatives of f, we set those partials equal to zero and solve the equations. We obtain an unexpected "**RootOf**" in our first solution, so the example shows how to resolve that. The notation "**s[1]**" refers to the first set of brackets in the solution "**s**" and "**allvalues**" yields the two solutions where "**RootOf**" has occurred. At a point, such as (x_0, y_0), where both partial derivatives are zero we test to see if the determinant of the Hessian $H(x,y) = \begin{bmatrix} f_{xx} & f_{xy} \\ f_{yx} & f_{yy} \end{bmatrix}$ evaluated at (x_0, y_0) is positive or negative. The positive case yields a relative maximum or minimum, while the negative case indicates that (x_0, y_0) is a saddle point. We test further if the determinant of the Hessian is positive because this can only occur if the second partial derivatives, f_{xx} and f_{yy}, have the same sign and may be interpreted as indicating concave "up" or "down". Because the entry of the Hessian, H, in the $(1,1)$ position is f_{xx}, we simply evaluate $H[1,1]$ at the point in question and we have our answer. We will need three packages for Maple: student, linalg, and plots.

Example: Find all critical points of the function $f(x,y) = x^3 + 3xy^2 - 4y^3 - 15x$ and determine which, if any, are relative maxima or minima or saddle points. For simplicity, we will define f as an expression.

```
> restart: with(student): with(plots): with(linalg):
> f:=x^3+3*x*y^2-4*y^3-15*x;
```

$$f := x^3 + 3xy^2 - 4y^3 - 15x$$

```
> fx:=diff(f,x);
```

$$fx := 3x^2 + 3y^2 - 15$$

```
> fy:=diff(f,y);
```

$$fy := 6xy - 12y^2$$

Solve the simultaneous equations for x and y.

```
> s:=solve({fx=0,fy=0},{x,y});
```

$$s := \{y = 0, x = RootOf(\_Z^2 - 5)\}, \{y = 1, x = 2\}, \{x = -2, y = -1\}$$

```
> s1:=allvalues(s[1]);
```

$$s1 := \{y = 0, x = \sqrt{5}\}, \{y = 0, x = -\sqrt{5}\}$$

```
> H:=hessian(f,[x,y]);
```

$$H := \begin{bmatrix} 6x & 6y \\ 6y & 6x - 24y \end{bmatrix}$$

Insert the first of the solutions listed in "s1" into H. The use of "op" reminds Maple that H is a matrix and allows access to the components of H.

```
> H1:=subs(s1[1],op(H));
```

$$H1 := \begin{bmatrix} 6\sqrt{5} & 0 \\ 0 & 6\sqrt{5} \end{bmatrix}$$

```
> a1:=det(H1);
```

$$a1 := 180$$

```
> b1:=subs(s1[1],H[1,1]);
```

Testing the sign of f_{xx} or fxx at the point.

$$b1 := 6\sqrt{5}$$

Because the Hessian is positive and b1 is positive (indicating 'concave up'), there is a relative minimum at `s1[1]` $= (\sqrt{5}, 0)$.

```
> v1:=subs(s1[1],f);
```

$$v1 := -10\sqrt{5}$$

```
> s2:=s1[2];
```

$$s2 := \{y = 0, x = -\sqrt{5}\}$$

This line was to demonstrate what ' `s1[2]` ' would produce.

```
> H2:=subs(s2,op(H));
```

$$H2 := \begin{bmatrix} -6\sqrt{5} & 0 \\ 0 & -6\sqrt{5} \end{bmatrix}$$

```
> a2:=det(H2);
```

$$a2 := 180$$

```
> b2:=subs(s2,H[1,1]);
```

$$b2 := -6\sqrt{5}$$

Because the Hessian is positive at s2 and b2 is negative (indicating 'concave down'), there is a relative maximum at s2$=(\sqrt{5},0)$. Now we compute the value of f at s2.

```
> v2:=subs(s2,f);
```

$$v2 := 10\sqrt{5}$$

```
> H3:=subs(s[2],op(H);
```

$$H3 := \begin{bmatrix} 12 & 6 \\ 6 & -12 \end{bmatrix}$$

's[2]' is the second set listed of the original set, 's'.

```
> a3:=det(H3);
```

$$a3 = -180$$

Because the determinant of the Hessian is negative, we have a saddle point at s[2]$=(2,1)$.

```
> v3:=subs(s[2],f);
```

$$v3 := -20$$

```
> H4:=subs(s[3],op(H));
```

$$H4 := \begin{bmatrix} -12 & -6 \\ -6 & 12 \end{bmatrix}$$

```
> a4:=det(H4);
```

$$a4 := -180$$

```
> v4:=subs(s[3],f);
```

$$v4 := 20$$

Because the determinant of the Hessian is negative at s[3]$=(-2,-1)$, there is a saddle point there.
Now let's see what the graph of f looks like.

```
> plot3d(f,x=-6..6,y=-5..5,color=black);
```



```
> contourplot(f,x=-6..6,y=-4..4,contours=[-28,-24,-20,-16,-12,-8,-4,0,4,8,12,16,20,24,
28],color=black);
```



Check out the location on the contour plot of the four critical points we located above. See if they make sense as far as being relative maxima, minima, or as saddle points.

C3M6 Problems

Solve for relative maxima, minima, and saddle points by using MAPLE and by pencil and paper. Include a MAPLE contour plot, but scale it down so as to not waste paper.

1. $f(x, y) = 2x^3 + 6xy^2 + 3y^3 - 150x + 4$
2. $f(x, y) = 4x^3 + 3xy^2 - y^3 - 24x + 4$